Math 2250−1
Friday September 2

Examples 4,5 from section 2.1 of the text, pages 83−84.

The Belgian demographer P.F. Verhulst introduced the logistic model around 1840, as a tool for studying human population growth. Our text demonstrates its superiority to the simple exponential growth model, and also illustrates why mathematical modelers must always exercise care, by comparing the two models to actual U.S. population data. here are actual U.S. populations by decade, from 1800−2000, see e.g. the table on page 84:

```
> restart : #clear Maple memory
> pops := [[1800, 5.3], [1810, 7.2], [1820, 9.6], [1830, 12.9],
      [1840, 17.1], [1850, 23.2], [1860, 31.4], [1870, 38.6],
      [1880, 50.2], [1890, 63.0], [1900, 76.2], [1910, 92.2],
      [1920, 106.0], [1930, 123.2], [1940, 132.2], [1950, 151.3],
      [1960, 179.3], [1970, 203.3], [1980, 225.6], [1990, 248.7],
      [2000, 281.4], [2010, 308.]] : #I added 2010 − between 306−313
> Digits := 5 : #the default is 8 significant digits, which will clutter up the formulas
```

Unlike Verhulst, the book uses data from 1800, 1850 and 1900 to get constants in our two models. We let t=0 correspond to 1800.

Exponential Model: For the exponential growth model $P(t) = P_0 e^{r t}$ we use the 1800 and 1900 data to get values for P_0 and r:

```
> P0 := 5.308;
  solve(P0 * exp(r * 100) = 76.212, r);
```

$$P0 := 5.308$$

$$0.026643 \quad (1)$$

```
> P1 := t → 5.308 * exp(.02664 * t);#exponential model −eqtn (9) page 83
```

$$P1 := t \rightarrow 5.308\, e^{0.02664\, t} \quad (2)$$

Logistic Model: We get P_0 from 1800, and use the 1850 and 1900 data to find k and M:

```
> P2 := t → M* P0 / (P0 + (M−P0) * exp(−M* k* t));
    #logisitic function, with our P0, eqtn (7) page 82
```

$$P2 := t \rightarrow \frac{M\, P0}{P0 + (M - P0)\, e^{-M k t}} \quad (3)$$

```
> solve({P2(50) = 23.192, P2(100) = 76.212}, {M, k});
```

$$\{M = 188.12, k = 0.00016772\} \quad (4)$$

```
>
```

```
> M := 188.12;
  k := .16772e-3;
  P2(t); #should be our logistic model function,
    #equation (11) page 84.
```

$$M := 188.12$$

$$k := 0.00016772$$

$$\frac{998.54}{5.308 + 182.81\, e^{-0.031551\, t}} \qquad (5)$$

Now compare the two models with the real data, and discuss:

```
> with(plots) :
  plot1 := plot(P1(t-1800), t = 1800 ..1950, color = black, linestyle = 3) :
   #this linestyle gives dashes for the exponential curve
  plot2 := plot(P2(t-1800), t = 1800 ..2010, color = black) :
  plot3 := pointplot( pops, symbol = cross) :
  display( {plot1, plot2, plot3}, title = ‘U.S. population data
  and models‘);
```



The exponential model takes no account of the fact that the U.S. has only finite resources. Any ideas on why the logistic model begins to fail (with our parameters) around 1950?